Problema da programação e roteamento da Mão de Obra – Implementação AMPL

Workforce Scheduling and Routing Problem (WSRP)

ARTIGO UTILIZADO:

Workforce Scheduling Linear Programming Formulation

T. Garaix**, M. Gondran*, P. Lacomme*, E. Mura***, N. Tchernev*

1) Université Clermont-Auvergne, LIMOS, UMR CNRS 6158, Campus des Cézeaux, 63178 Aubière Cedex France (e-mail: (gondran, placomme, tchernev}@isima.fr).

2) Ecole des Mines de Saint-Etienne, 158 cours Fauriel, 42000 Saint-Etienne

France (e-mail: garaix@emse.fr).

3) Université de Technologie de Troyes, ICD-LOSI, UMR CNRS 6281, 12 rue Marie Curie, CS 42060, 10004 Troyes Cedex, France (e-mail: enzy,mura@utt.fr).

Link do artigo: Workforce Scheduling Linear Programming Formulation

Problema de programação e roteamento HHC



PARÂMETROS:

- W - conjunto de trabalhadores
- T - conjunto de tarefas
- $dist_{i,j}$ - distância da tarefa i à tarefa j
- p_j^w - custo do trabalhador w para realizar j.
- Dur_j - tempo de processamento da tarefa j
- r_j - número de trabalhadores necessários para realizar a tarefa j
- pw_j^w - nível de satisfação ($pw_j^w \in [0,1]$) quando o trabalhador w é atribuído à tarefa j
- pa_j^w - nível de satisfação ($pa_j^w \in [0,1]$) quando o trabalhador w é atribuído à tarefa j considerando as preferências regionais (a tarefa j está localizada em uma região k)
- ps_j^w - nível de satisfação da habilidade ($ps_j^w \in [0,1]$) com $ps_j^w = \max(ps_j^i)$ quando a tarefa j é atribuída ao trabalhador w
- p_j^w = $(pa_j^w + pw_j^w + ps_j^w) \forall w \in W, j \in T$ - é a qualidade do serviço.
- $[e_j, l_j]$ - janela de tempo da tarefa j
- $[TW_{inf}^w, TW_{sup}^w]$ - janela de tempo de trabalho do trabalhador w

PARÂMETROS:

```
# PARÂMETROS:
param M := 99999999999999999999999999999999; # Big M
param nT := 6; # Número total de tarefas: T
param nW := 3; # Número total de trabalhadores
set T := 0..nT; # Conjunto de tarefas: T = {t1, ti, ..., tT}
set W := 1..nW; # Conjunto de trabalhadores: W = {w1, wi, ..., wT}
set A within T cross T; # Conjunto de arestas
# Definimos um grafo G = (V,E)
# Onde V é o conjunto de nós e E o conjunto de arestas
param dist{(i,j) in A}; # Distância da tarefa i à tarefa j
param p{T,W}; # Custo do trabalhador w para realizar j
param Dur{j in T}; # Tempo de processamento da tarefa j
param r{j in T}; # Número de trabalhadores necessários para realizar a tarefa j
param pw{j in T, w in W} >= 0, <= 1; # Nível de satisfação ((pw)j^w ∈ [0,1]) quando o trabalhador w é atribuído à tarefa j
param pa{j in T, w in W} >= 0, <= 1; # Nível de satisfação ((pa)w^j ∈ [0,1]) quando o trabalhador w é atribuído à tarefa j considera
param ps{j in T, w in W} >= 0, <= 1; # Nível de satisfação da habilidade ((ps)w^j ∈ [0,1]) com (ps)w^j=max((ps)i^j) quando a tarefa
param rojw{j in T, w in W}; #= pa[j,w] + pw[j,w] + ps[j,w]; # Qualidade do serviço = pa + pw + ps
param ej{j in T}; # Janela de tempo da tarefa j -> Limite inferior
param lj{j in T}; # Janela de tempo da tarefa j -> Limite superior
param TWinf{w in W}; # Janela de tempo de trabalho do trabalhador w -> Limite inferior
param TWsup{w in W}; # janela de tempo de trabalho do trabalhador w -> Limite superior
# Ordem de prioridade dos pesos: λ4 ⩾ λ3 ⩾ λ2 ⩾ λ1
param lambda1 := 0.4; # Peso 1
param lambda2 := 0.65; # Peso 2
param lambda3 := 0.75; # Peso 3
param lambda4 := 1.0; # Peso 4
```

VARIÁVEIS BINÁRIAS:

- $x_{i,j}^{w} = \begin{cases} 1, & \text{se o trabalhador } w \text{ passar da tarefa } i \text{ para } j \\ 0, & \text{senão} \end{cases}$
- $x_{0,j}^{w} = \begin{cases} 1, & \text{se o trabalhador } w \text{ se deslocar do local de partida para a tarefa } i \\ 0, & \text{senão} \end{cases}$
- $x_{i,0}^{w} = \begin{cases} 1, & \text{se o trabalhador } w \text{ passar da tarefa } i \text{ para o local de término } 0 \\ 0, & \text{senão} \end{cases}$
- $\psi_{j}^{w} = \begin{cases} 1, & \text{se o trabalhador } w \text{ for designado para uma tarefa } j \text{ situada fora da região de preferência} \\ 0, & \text{senão} \end{cases}$
- $\theta_{j}^{w} = \begin{cases} 1, & \text{se a violação da janela de tempo ocorrer quando a tarefa } j \text{ for atribuída ao trabalhador } w \\ 0, & \text{senão} \end{cases}$

VARIÁVEIS CONTÍNUAS:

- y_j - número de trabalhadores não disponíveis para realizar a tarefa j
- t_j - hora de início da tarefa j
- d_j^w - hora de saída do trabalhador w da tarefa j
- a_j^w - hora de chegada do trabalhador w à tarefa j

VARIÁVEIS:

```
# VARIÁVEIS DE DECISÃO:
## Binárias:
var x{(i,j) in A, w in W}, binary; # x[i,j,w] = 1 se o trabalhador w passar da tarefa i para j
var psi{j in T, w in W}, binary; # psi[j] = 1 se o trabalhador w for designado para uma tarefa
var theta{j in T, w in W}, binary; # theta[j] = 1 se a violação da janela de tempo ocorrer quan
var c{j in T, w in W}, binary; # c[j,w] = 1 se o trabalhador tiver sido incluído no contrato do

## Contínuas:
var y{j in T} integer >= 0; # Número de trabalhadores não disponíveis para realizar a tarefa j
var t{j in T} >= 0; # Hora de início da tarefa j
var d{T,W} >= 0; # Hora de saída do trabalhador w da tarefa j
var a{T,W} >= 0; # Hora de chegada do trabalhador w à tarefa j
```

FUNÇÃO OBJETIVO:

A função objetivo a ser minimizada envolve quatro critérios que são balanceados por

$$\min f(s) = \lambda_1 \sum_{w=1}^{W} \sum_{i=0}^{T} \sum_{j=1}^{T} (dist_{i,j} + p_j^w) x_{i,j}^w + \lambda_2 \sum_{j=1}^{T} (3r_j - \sum_{i=0}^{T} \sum_{w=1}^{W} p_j^w x_{i,j}^w) + \lambda_3 \sum_{j=1}^{T} \sum_{w=1}^{W} (\psi_j^w + \theta_j^w) + \lambda_4 \sum_{j=1}^{T} y_j$$

quatro pesos (λ) correspondentes aos níveis de prioridade.

Ordem de prioridade dos pesos: $\lambda_4 \succcurlyeq \lambda_3 \succcurlyeq \lambda_2 \succcurlyeq \lambda_1$

FUNÇÃO OBJETIVO:

```
# FUNÇÃO OBJETIVO:
minimize Objective:
lambda1 * sum{w in W, i in T, j in T: j > 0} (dist[i,j] + p[j,w]) * x[i,j,w] + # Critério 1
lambda2 * sum{j in T: j > 0} (3 * r[j] - sum{i in T, w in W} rojw[i,w] * x[i,j,w]) + # Critério 2
lambda3 * sum{j in T, w in W: j > 0} (psi[j,w] + theta[j,w]) + # Critério 3
lambda4 * sum{j in T: j > 0} y[j] # Critério 4
;
```

CONJUNTO DE RESTRIÇÕES 1:

A viagem de um trabalhador w da tarefa i para a tarefa j é definida por $x^w_{i,j} = 1$ (tarefa 0 é o local inicial e tarefa $|T|$ é o local final). Assim, estas restrições garantem que para um trabalhador w: de uma tarefa i um trabalhador w pode viajar para no máximo uma outra tarefa (1ª restrição); para uma tarefa j um trabalhador pode viajar de no máximo uma tarefa (2ª restrição); e se w for atribuído a uma tarefa j após a tarefa i então $x^w_{i,j} = 1$ e (3ª restrição) pode ser reescrita como $\sum_{u=0}^{T} x^w_{j,u} = 1$ significando que w deve viajar para outra tarefa (ou para a localização final) e não pode ficar nesta posição.

- $\sum_{j=0}^{T} x^w_{i,j} \preccurlyeq 1, \forall i = 0..|T|, w = 1..|W|$
- $\sum_{i=0}^{T} x^w_{i,j} \preccurlyeq 1, \forall i = 0..|T|, w = 1..|W|$
- $\sum_{i=0}^{T} x^w_{i,j} = \sum_{u=0}^{T} x^w_{j,u'} \forall i = 0..|T|, w = 1..|W|$

CONJUNTO DE RESTRIÇÕES 1:

```
# RESTRIÇÕES:
## Conjunto 1:
Constraint_1{i in T, w in W}: sum{j in T: (i,j) in A} x[i,j,w] <= 1;
Constraint_2{j in T, w in W}: sum{i in T: (i,j) in A} x[i,j,w] <= 1;
Constraint_3{j in T, w in W}: sum{i in T: (i,j) in A} x[i,j,w] = sum{u in T: (j,u) in A} x[j,u,w];
```

CONJUNTO DE RESTRIÇÕES 2:

O número de trabalhadores atribuídos à tarefa j deve estar de acordo com r_j ou seja, com o número de trabalhadores necessários para realizar a tarefa j. y_j é o número de trabalhadores ausentes para a tarefa alcançada j e é definida por: $y_j = r_j - \sum_{w=1}^{W} \sum_{i=0}^{T} x_{i,j}^{w}$. Se uma tarefa j exigir três trabalhadores ($r_j = 3$), mas apenas um trabalhador w_1 é atribuído a j para que exista um e apenas um $x_{i,j}^{w} = 1$ e a restrição poderá ser reescrita como $\sum_{w=1}^{W} \sum_{i=0}^{T} x_{i,j}^{w} + y_j = 1 + y_i = 3$. Por consequência, $y_i = 2$ que é o número de trabalhadores ausentes para realizar a tarefa.

- $\sum_{w=1}^{W} \sum_{i=0}^{T} x_{i,j}^{w} + y_j = r_j, \forall j = 1..|T|$

CONJUNTO DE RESTRIÇÕES 2:

```
## Conjunto 2:
Constraint_4{j in T: j > 0}: sum{w in W, i in T} x[i,j,w] + y[j] = r[j];
```

CONJUNTO DE RESTRIÇÕES 3:

- $\psi_j^w + M.pa_j^w \succcurlyeq \sum_{i=0}^{T} x_{i,j}^w, \forall w = 1..|W|, j = 1..|T|$
- $M.\theta_j \succcurlyeq t_j + Dur_j - TW_{sup}^w + (\sum_{i=0}^{T} x_{i,j}^w - 1).M, \forall w = 1..|W|, j = 1..|T|$
- $M.\theta_j \succcurlyeq TW_{inf}^w - t_j + (\sum_{i=0}^{T} x_{i,j}^w - 1).M, \forall w = 1..|W|, j = 1..|T|$

CONJUNTO DE RESTRIÇÕES 3:

Idealmente, um trabalhador deve ser designado para tarefas em suas regiões geográficas disponíveis. No entanto, as violações da região de preferência são possíveis e devem ser medidas para definir a terceira parte da função objetivo. Na restrição 1, se $pa_j^w = 0$, o trabalhador w é atribuído à tarefa j que está localizado em uma região não desejada, a restrição 1 pode ser reescrita como, $\psi_j^w \succcurlyeq \sum_{i=0}^{T} x_{i,j}^w$ implicando $\psi_j^w \succcurlyeq 1$.

Quando ambas as tarefas i e j são atribuídas ao trabalhador w, $x_{i,j}^w = 1$ e a restrição 2:

$$M.\theta_j \succcurlyeq t_j + Dur_j - TW_{sup}^w + (\sum_{i=0}^{T} x_{i,j}^w - 1).M$$

$$M.\theta_j \succcurlyeq TW_{inf}^w - t_j + (\sum_{i=0}^{T} x_{i,j}^w - 1).M$$

pode ser reescrita como:

$$M.\theta_j \succcurlyeq t_j + Dur_j - TW_{sup}^w$$

$$M.\theta_j \succcurlyeq TW_{inf}^w - t_j$$

A restrição 2 garante que, se o tempo de conclusão $(t_j + Dur_j)$ da tarefa j exceder TW_{sup}^w, a variável binária θ_j é definida como 1. Como $t_j + Dur_j - TW_{sup}^w > 0$, a restrição 2 é reescrita como $M.\theta_j \succcurlyeq t_j + Dur_j - TW_{sup}^w > 0$ e garante que $\theta_j = 1$. A restrição 3 garante que $\theta_j = 1$, se $t_j < TW_{inf}^w$, isto é, quando $TW_{inf}^w - t_j > 0$, o que significa que o trabalhador w tem que executar tarefa j antes do final de sua janela de tempo de trabalho.

CONJUNTO DE RESTRIÇÕES 3:

```
## Conjunto 3:
Constraint_5{j in T, w in W: j > 0}: (psi[j,w] + M * pa[j,w]) >= sum{i in T} x[i,j,w];
Constraint_6{j in T, w in W: j > 0}: M * theta[j,w] >= sum{i in T} (x[i,j,w] - 1) * M + t[j] + Dur[j] - TWsup[w];
Constraint_7{j in T, w in W: j > 0}: M * theta[j,w] >= sum{i in T} (x[i,j,w] - 1) * M - t[j] + TWinf[w];
```

CONJUNTO DE RESTRIÇÕES 4:

Nos cenários abordados nesta apresentação, um trabalhador pode realizar uma tarefa exigida por um cliente se e somente se o trabalhador tiver sido incluído no contrato do cliente (restrição abaixo), neste caso $c_j^w = 1$, caso contrário $c_j^w = 0$ e $\sum_{i=0}^{T} x_{i,j}^w < 0$ que impõe $x_{i,j}^w = 0$.

- $\sum_{i=0}^{T} x_{i,j}^w \preccurlyeq c_j^w, \forall j = 1..|T|, w = 1..|W|$

CONJUNTO DE RESTRIÇÕES 4:

```
## Conjunto 4:
Constraint_8{j in T, w in W: j > 0}: c[j,w] >= sum{i in T} x[i,j,w];

```

CONJUNTO DE RESTRIÇÕES 5:

Se um trabalhador w for atribuído a uma tarefa j após uma tarefa i, então $x^w_{i,j} = 1$, então a restrição 1 pode ser reescrita como $d^w_j \succcurlyeq (t_j + Dur_j)$ implicando que o tempo de partida d^w_j do trabalhador w da tarefa j deve ser maior que o tempo de início t_j de j mais o tempo de processamento Dur_j de j. Se $x^w_{i,j} = 0$, (restrição 1) pode ser reescrito como: $d^w_j \succcurlyeq -M$ e essa restrição é válida.

A restrição 2 garante que, se um trabalhador w for atribuído à tarefa j após a tarefa i ($x^w_{i,j} = 1$) então $a^w_j \succcurlyeq (d^w_{s_i} + dist_{i,j})$ deve manter, implicando que o tempo de chegada a^w_j de w na tarefa j está depois de seu horário de partida d^w_i da tarefa i mais a distância entre as tarefas i e j. Se $x^w_{i,j} = 0$, (restrição 2) pode ser reescrito como: $a^w_j \succcurlyeq -M$ e esta restrição é válida.

- $d^w_j \succcurlyeq (t_j + Dur_j) + (\sum_{i=0}^{T} x^w_{i,j} - 1).\, M, \forall w = 1..|W|, j = 1..|T|$
- $a^w_j \succcurlyeq (d^w_i + dist_{i,j}) + (x^w_{i,j} - 1).\, M, \forall w = 1..|W|, i = 0..|T|, j = 1..|T|$

CONJUNTO DE RESTRIÇÕES 5:

```
## Conjunto 5:
Constraint_9{j in T, w in W: j > 0}: d[j,w] >= sum{i in T} (x[i,j,w] - 1) * M + (t[j] + Dur[j]);
Constraint_10{i in T, j in T, w in W: j > 0}: a[j,w] >= (x[i,j,w] - 1) * M + (d[i,w] + dist[i,j]);
```

CONJUNTO DE RESTRIÇÕES 6:

A hora de início t_j da tarefa j deve ser posterior à hora de chegada a_j^w de todos os trabalhadores necessários w para realizar a tarefa (restrição 1). Além disso, uma tarefa j tem uma janela de tempo $[e_j , l_j]$ e pode começar somente após o início e_j de sua janela de tempo (restrição 2) e antes do final de sua janela de tempo (restrição 3).

- $t_j \succcurlyeq a_j^w, \forall w = 1..|W|, j = 1..|T|$
- $t_j \succcurlyeq e_j, \forall j = 1..|T|$
- $t_j \preccurlyeq l_j, \forall j = 1..|T|$

CONJUNTO DE RESTRIÇÕES 6:

```
## Conjunto 6:
Constraint_11{j in T, w in W: j > 0}: a[j,w] <= t[j];
Constraint_12{j in T: j > 0}: ej[j] <= t[j];
Constraint_13{j in T: j > 0}: lj[j] <= t[j];
```

CONJUNTO DE RESTRIÇÕES - DOMÍNIO DAS VARS:

$$x_{i,j}^{w}, \theta_j^{w}, \psi_j^{w} \in \{0,1\}, y_j \in \mathbb{N}, t_i, a_j^{w}, d_j^{w} \in \mathbb{R}$$